

Identification Division.

Program-ID. J7200552.

```
*
* Determine the current execution environment - are we
* running in CICS, Batch, or Unix?
*
* Note that the run environment is cached in Working-Storage
* so that multiple calls in a single run unit (a batch job or
* CICS transaction) will only result in one call to J7200551.
*
* Syntax:
* Working-Storage Section.
* 77 CICS-OR-WHAT PIC X(008) VALUE 'J7200552'.
* 77 CICS-OR-WHAT-SW PIC X(001) VALUE SPACE.
* 88 IN-CICS VALUE 'C'.
* 88 IN-BATCH VALUE 'B'.
* 88 IN-UNIX VALUE 'U'.
*
* Procedure Division.
* CALL CICS-OR-WHAT USING CICS-OR-WHAT-SW END-CALL
* IF RETURN-CODE = 0
* EVALUATE TRUE
* WHEN IN-CICS
* do cics things
* WHEN IN-BATCH
* do batch things
* WHEN IN-UNIX
* DISPLAY 'How did I get here?'
* END-EVALUATE
* ELSE
* error handling
* END-IF
*
* Return codes:
* 0 - Success
* 8 - Unable to determine environment - LE error
* 12 - Unable to determine environment - unknown cause
*
* Check SYSOUT in batch, CEEMSG (the CESE TD queue) in CICS,
* stderr or stdout in Unix for LE error messages in the event
* of a bad return code.
*
```

Environment Division.

Data Division.

Working-Storage Section.

```
01 CONSTANTS.
05 MYNAME PIC X(008) VALUE 'J7200552'.
05 INFO-PGM PIC X(008) VALUE 'J7200551'.
```

```
01 SWITCHES.
05 FIRST-TIME-SW PIC X(001) VALUE SPACE.
88 FIRST-TIME-THRU VALUE SPACE.
```

Local-Storage Section.

```
01 WS-SYS-ENV.
05 WS-SYS PIC X(032) VALUE SPACES.
05 WS-ENV PIC X(032) VALUE SPACES.
05 WS-LANG PIC X(032) VALUE SPACES.
```

```
01 WS-LE-INFO.
05 WS-LE-PRDT-NB PIC 9(005) VALUE ZEROES.
05 WS-LE-VERS PIC 9(005) VALUE ZEROES.
05 WS-LE-RLSE PIC 9(005) VALUE ZEROES.
05 WS-LE-MDFN PIC 9(005) VALUE ZEROES.
```

Linkage Section.

```
01 RUN-ENV PIC X(001).
88 RUN-ENV-CICS VALUE 'C'.
88 RUN-ENV-BATCH VALUE 'B'.
88 RUN-ENV-UNIX VALUE 'U'.
88 RUN-ENV-UNKNOWN VALUE SPACE.
```

Procedure Division Using RUN-ENV.

```
IF FIRST-TIME-THRU
*   Get information about our environment
   CALL INFO-PGM USING
     WS-SYS-ENV
     WS-LE-INFO
   END-CALL
   IF RETURN-CODE = 0
     EVALUATE TRUE
       WHEN WS-SYS(1:1) = '1'
         WHEN WS-SYS(2:1) = '1'
         WHEN WS-SYS(3:1) = '1'
         WHEN WS-SYS(4:1) = '1'
           SET RUN-ENV-CICS TO TRUE
       WHEN WS-SYS(6:1) = '1'
         SET RUN-ENV-BATCH TO TRUE
       WHEN WS-SYS(7:1) = '1'
         SET RUN-ENV-UNIX TO TRUE
```

```
        WHEN OTHER
            SET RUN-ENV-UNKNOWN TO TRUE
            MOVE 12 TO RETURN-CODE
        END-EVALUATE
    ELSE
        SET RUN-ENV-UNKNOWN TO TRUE
        MOVE 8 TO RETURN-CODE
    END-IF
    MOVE RUN-ENV TO FIRST-TIME-SW
ELSE
    MOVE FIRST-TIME-SW TO RUN-ENV
END-IF

GOBACK.
```